# Appendix A

Maple code for Examples 1-4 from "Extending Explicit and Linealry Implicit ODE Solvers for Index-1 DAEs", M. T. Lawder,
V. Ramadesigan, B. Suthar and V. R. Subramanian, Computers and Chemical Engineering, in press (2015).

Use y1, y2, etc. for all differential variables and z1, z2, etc. for all algebraic variables

**Example 1**

```
> restart;
> with(plots):
```

Enter all ODEs in eqode

```
> eqode:=[diff(y1(t),t)=-y1(t)^2+z1(t)];
```

$$\text{eqode} := \left[\frac{d}{dt}\,y1(t) = -y1(t)^2 + z1(t)\right]$$

Enter all AEs in eqae

```
> eqae:=[cos(y1(t))-z1(t)^0.5=0];
```

$$\text{eqae} := [\cos(y1(t)) - z1(t)^{0.5} = 0]$$

Enter all initial conditions for differential variables in icodes

```
> icodes:=[y1(0)=0.25];
```

$$\text{icodes} := [y1(0) = 0.25]$$

Enter all intial conditions for algebraic variables in icaes

```
> icaes:=[z1(0)=0.8];
```

$$\text{icaes} := [z1(0) = 0.8]$$

Enter parameters for perturbation value (mu), switch function (q and tint), and runtime (tf)

```
> pars:=[mu=0.1,q=1000,tint=1,tf=5];
```

$$\text{pars} := [\mu = 0.1, q = 1000, \text{tint} = 1, \text{tf} = 5]$$

Choose solving method (1 for explicit, 2 for implicit)

```
> Xexplicit:=2;
```

$$\text{Xexplicit} := 2$$

Standard solver requires IC z(0)=0.938791 or else it will fail

```
> solx:=dsolve({eqode[1],eqae[1],icodes[1],icaes[1]},numeric):
Error, (in dsolve/numeric/DAE/checkconstraints) the initial conditions do not
satisfy the algebraic constraints
  error = .745e-1, tolerance = .559e-6, constraint =
cos(y1(t))-z1(t)^.50000000000000000000
```

```
> ff:=subs(pars,1/2+1/2*tanh(q*(t-tint)));
```

$$\text{ff} := \frac{1}{2} + \frac{1}{2}\tanh(1000\,t - 1000)$$

```
> NODE:=nops(eqode);NAE:=nops(eqae);
```

$$\text{NODE} := 1$$

$$\text{NAE} := 1$$

```
> for XX from 1 to NODE do
  EQODE||XX:=lhs(eqode[XX])=rhs(eqode[XX])*ff;
  end do;
```

$$\text{EQODE1} := \frac{d}{dt}\,y1(t) = (-y1(t)^2 + z1(t))\left(\frac{1}{2} + \frac{1}{2}\tanh(1000\,t - 1000)\right)$$

```
> for XX from 1 to NAE do
  EQAE||XX:=subs(pars,-mu*(diff(rhs(eqae[XX])-lhs(eqae[XX]),t))=rhs(
  eqae[XX])-lhs(eqae[XX]));
  end do;
```

$$\text{EQAE1} := -0.1\,\sin(y1(t))\left(\frac{d}{dt}\,y1(t)\right) - \frac{0.05\left(\frac{d}{dt}\,z1(t)\right)}{z1(t)^{0.5}} = -\cos(y1(t)) + z1(t)^{0.5}$$

```
>
> Dvars1:={seq(diff(z||x(t),t)=D||x,x=1..NAE)};
```

$$\text{Dvars1} := \left\{\frac{d}{dt}\,z1(t) = D1\right\}$$

```
> Dvars2:={seq(rhs(Dvars1[x])=lhs(Dvars1[x]),x=1..NAE)};
```

$$\text{Dvars2} := \left\{D1 = \frac{d}{dt}\,z1(t)\right\}$$

```
> icsn:=seq(subs(y||x(0)=y||x(t),icodes[x]),x=1..NODE),seq(subs(z||x
  (0)=z||x(t),icaes[x]),x=1..NAE);
```

$$\text{icsn} := y1(t) = 0.25,\ z1(t) = 0.8$$

```
> for j from 1 to NAE do
> EQAEX||j:=subs(Dvars1,eqode,icsn,Dvars2,lhs(EQAE||j))=rhs(EQAE||j)
  ;
> end do:
> Sys:={seq(EQODE||x,x=1..NODE),seq(EQAEX||x,x=1..NAE),seq(icodes[x]
  ,x=1..NODE),seq(icaes[x],x=1..NAE)};
```

$$\text{Sys} := \left\{-0.01824604200 - 0.05590169945\left(\frac{d}{dt}\,z1(t)\right) = -\cos(y1(t)) + z1(t)^{0.5},\ y1(0) = 0.25,\right.$$
$$\left. z1(0) = 0.8,\ \frac{d}{dt}\,y1(t) = (-y1(t)^2 + z1(t))\left(\frac{1}{2} + \frac{1}{2}\tanh(1000\,t - 1000)\right)\right\}$$

```
> if Xexplicit=1 then
> sol:=dsolve(Sys,numeric,initstep=0.1,compile=true):
> else
> sol:=dsolve(Sys,numeric,stiff=true,implicit=true,initstep=0.1,comp
  ile=true):
  end if:
>
> for XX from 1 to NODE do
  a||XX:=odeplot(sol,[t,y||XX(t)],0..subs(pars,tf),color=red);
```

```
  end do:
> for XX from NODE+1 to NODE+NAE do
  a||XX:=odeplot(sol,[t,z||(XX-NODE)(t)],0..subs(pars,tf),color=blue
  );
  end do:
> display(seq(a||x,x=1..NODE+NAE),axes=boxed);
```




End Example 1

>

**Example 2**

```
> restart;
> with(plots):
> eq1:=diff(y1(t),t)=j1*W/F/rho/V;
```

$$eq1 := \frac{d}{dt} y1(t) = \frac{j1\ W}{F\ \rho\ V}$$

```
> eq2:=j1+j2=iapp;
```

$$eq2 := j1 + j2 = iapp$$

```
> j1:=io1*(2*(1-y1(t))*exp((0.5*F/R/T)*(z1(t)-phi1))-2*y1(t)*exp((-0
  .5*F/R/T)*(z1(t)-phi1)));
```

$$j1 := io1 \left( 2\,(1 - y1(t))\, e^{\left( \frac{0.5\ F\ (z1(t) - \phi1)}{R\ T} \right)} - 2\ y1(t)\, e^{\left( -\frac{0.5\ F\ (z1(t) - \phi1)}{R\ T} \right)} \right)$$

```
> j2:=io2*(exp((F/R/T)*(z1(t)-phi2))-exp((-F/R/T)*(z1(t)-phi2)));
```

$$j2 := io2 \left( e^{\left( \frac{F\ (z1(t) - \phi2)}{R\ T} \right)} - e^{\left( -\frac{F\ (z1(t) - \phi2)}{R\ T} \right)} \right)$$

```
> params:={F=96487,R=8.314,T=298.15,phi1=0.420,phi2=0.303,W=92.7,V=1
  e-5,io1=1e-4,io2=1e-10,iapp=1e-5,rho=3.4};
```

params := {F = 96487, R = 8.314, T = 298.15, V = 0.00001, W = 92.7, io1 = 0.0001,

```
    io2 = 0.1 10^-9, ρ = 3.4, iapp = 0.00001, ϕ1 = 0.420, ϕ2 = 0.303 }
> eqode:=[subs(params,eq1)];
```

$$\text{eqode} := \left[ \frac{d}{dt} y1(t) = 0.005651477584\,(1 - y1(t))\, e^{(19.46229155\, z1(t) - 8.174162450)} - 0.005651477584\, y1(t)\, e^{(-19.46229155\, z1(t) + 8.174162450)} \right]$$

```
> eqae:=[subs(params,eq2)];
```

$$\text{eqae} := [\, 0.0002\,(1 - y1(t))\, e^{(19.46229155\, z1(t) - 8.174162450)} - 0.0002\, y1(t)\, e^{(-19.46229155\, z1(t) + 8.174162450)} + 0.1\, 10^{-9}\, e^{(38.92458310\, z1(t) - 11.79414868)} - 0.1\, 10^{-9}\, e^{(-38.92458310\, z1(t) + 11.79414868)} = 0.00001 \,]$$

```
> icodes:=[y1(0)=0.05];
```

$$\text{icodes} := [\, y1(0) = 0.05 \,]$$

```
> icaes:=[z1(0)=0.7];
```

$$\text{icaes} := [\, z1(0) = 0.7 \,]$$

```
> solx:=dsolve({eqode[1],eqae[1],icodes[1],icaes[1]},type=numeric):
Error, (in dsolve/numeric/DAE/checkconstraints) the initial conditions do not
satisfy the algebraic constraints
  error = .447e9, tolerance = .880e4, constraint =
-2000000*(-1+y1(t))*exp(19.462291550000000000*z1(t)-8.174162450000000000)-200
0000*y1(t)*exp(-19.462291550000000000*z1(t)+8.174162450000000000)+exp(38.9245
8310000000000*z1(t)-11.794148680000000000)-exp(-38.924583100000000000*z1(t)
+11.794148680000000000)-100000

> pars:=[mu=0.00001,q=1000,tint=1,tf=5001];
```

$$\text{pars} := [\, \mu = 0.00001,\ q = 1000,\ \text{tint} = 1,\ \text{tf} = 5001 \,]$$

```
> Xexplicit:=2;
```

$$\text{Xexplicit} := 2$$

```
> ff:=subs(pars,1/2+1/2*tanh(q*(t-tint)));
```

$$\text{ff} := \frac{1}{2} + \frac{1}{2} \tanh(1000\, t - 1000)$$

```
> NODE:=nops(eqode):NAE:=nops(eqae);
```

$$\text{NAE} := 1$$

```
> for XX from 1 to NODE do
  EQODE||XX:=lhs(eqode[XX])=rhs(eqode[XX])*ff:
  end do;
```

$$\text{EQODE1} := \frac{d}{dt} y1(t) = (0.005651477584\,(1 - y1(t))\, e^{(19.46229155\, z1(t) - 8.174162450)} - 0.005651477584\, y1(t)\, e^{(-19.46229155\, z1(t) + 8.174162450)})\left(\frac{1}{2} + \frac{1}{2} \tanh(1000\, t - 1000)\right)$$

```
> for XX from 1 to NAE do
```

```
EQAE||XX:=subs(pars,-mu*(diff(rhs(eqae[XX])-lhs(eqae[XX]),t))=rhs(
eqae[XX])-lhs(eqae[XX]));
end do;
```

$$\begin{aligned}
EQAE1 := {} & -0.2\,10^{-8}\left(\frac{d}{dt}y1(t)\right)e^{(19.46229155\,z1(t)-8.174162450)} \\
& + 0.3892458310\,10^{-7}\,(1-y1(t))\left(\frac{d}{dt}z1(t)\right)e^{(19.46229155\,z1(t)-8.174162450)} \\
& - 0.2\,10^{-8}\left(\frac{d}{dt}y1(t)\right)e^{(-19.46229155\,z1(t)+8.174162450)} \\
& + 0.3892458310\,10^{-7}\,y1(t)\left(\frac{d}{dt}z1(t)\right)e^{(-19.46229155\,z1(t)+8.174162450)} \\
& + 0.3892458310\,10^{-13}\left(\frac{d}{dt}z1(t)\right)e^{(38.92458310\,z1(t)-11.79414868)} \\
& + 0.3892458310\,10^{-13}\left(\frac{d}{dt}z1(t)\right)e^{(-38.92458310\,z1(t)+11.79414868)} = 0.00001 \\
& - 0.0002\,(1-y1(t))\,e^{(19.46229155\,z1(t)-8.174162450)} \\
& + 0.0002\,y1(t)\,e^{(-19.46229155\,z1(t)+8.174162450)} \\
& - 0.1\,10^{-9}\,e^{(38.92458310\,z1(t)-11.79414868)} \\
& + 0.1\,10^{-9}\,e^{(-38.92458310\,z1(t)+11.79414868)}
\end{aligned}$$

```
> Dvars1:={seq(diff(z||x(t),t)=D||x,x=1..NAE)};
```

$$Dvars1 := \left\{\frac{d}{dt}z1(t) = D1\right\}$$

```
> Dvars2:={seq(rhs(Dvars1[x])=lhs(Dvars1[x]),x=1..NAE)};
```

$$Dvars2 := \left\{D1 = \frac{d}{dt}z1(t)\right\}$$

```
> icsn:=seq(subs(y||x(0)=y||x(t),icodes[x]),x=1..NODE),seq(subs(z||x
(0)=z||x(t),icaes[x]),x=1..NAE);
```

$$icsn := y1(t) = 0.05,\ z1(t) = 0.7$$

```
> for j from 1 to NAE do
> EQAEX||j:=subs(Dvars1,eqode,icsn,Dvars2,lhs(EQAE||j))=rhs(EQAE||j)
;
> end do;
```

$$\begin{aligned}
EQAEX1 := {} & -0.2\,10^{-8}\,(0.005368903705\,e^{5.449441630} - 0.0002825738792\,e^{(-5.449441630)})\,e^{5.449441630} \\
& + 0.3697835394\,10^{-7}\left(\frac{d}{dt}z1(t)\right)e^{5.449441630} - 0.2\,10^{-8} \\
& (0.005368903705\,e^{5.449441630} - 0.0002825738792\,e^{(-5.449441630)})\,e^{(-5.449441630)}
\end{aligned}$$

$$+ 0.1946229155\ 10^{-8}\left(\frac{d}{dt}z1(t)\right)e^{(-5.449441630)}$$
$$+ 0.3892458310\ 10^{-13}\left(\frac{d}{dt}z1(t)\right)e^{15.45305949}$$
$$+ 0.3892458310\ 10^{-13}\left(\frac{d}{dt}z1(t)\right)e^{(-15.45305949)} = 0.00001$$
$$- 0.0002\ (1 - y1(t))\ e^{(19.46229155\ z1(t) - 8.174162450)}$$
$$+ 0.0002\ y1(t)\ e^{(-19.46229155\ z1(t) + 8.174162450)}$$
$$- 0.1\ 10^{-9}\ e^{(38.92458310\ z1(t) - 11.79414868)}$$
$$+ 0.1\ 10^{-9}\ e^{(-38.92458310\ z1(t) + 11.79414868)}$$

```
> Sys:={seq(EQODE||x,x=1..NODE),seq(EQAEX||x,x=1..NAE),seq(icodes[x]
  ,x=1..NODE),seq(icaes[x],x=1..NAE)};
```

$$Sys := \{-0.5810962488\ 10^{-6} + 0.8802389238\ 10^{-5}\left(\frac{d}{dt}z1(t)\right) = 0.00001$$
$$- 0.0002\ (1 - y1(t))\ e^{(19.46229155\ z1(t) - 8.174162450)}$$
$$+ 0.0002\ y1(t)\ e^{(-19.46229155\ z1(t) + 8.174162450)}$$
$$- 0.1\ 10^{-9}\ e^{(38.92458310\ z1(t) - 11.79414868)}$$
$$+ 0.1\ 10^{-9}\ e^{(-38.92458310\ z1(t) + 11.79414868)}, y1(0) = 0.05, z1(0) = 0.7, \frac{d}{dt}y1(t) = ($$
$$0.005651477584\ (1 - y1(t))\ e^{(19.46229155\ z1(t) - 8.174162450)}$$
$$- 0.005651477584\ y1(t)\ e^{(-19.46229155\ z1(t) + 8.174162450)})\left(\frac{1}{2} + \frac{1}{2}\tanh(1000\ t - 1000)\right)\}$$

```
> if Xexplicit=1 then
> sol:=dsolve(Sys,numeric,maxfun=0):
> else
> sol:=dsolve(Sys,numeric,stiff=true,implicit=true,maxfun=0):
> end if:
>
> for XX from 1 to NODE do
  a||XX:=odeplot(sol,[t,y||XX(t)],0..subs(pars,tf),color=red);
  end do:
> for XX from NODE+1 to NODE+NAE do
  a||XX:=odeplot(sol,[t,z||(XX-NODE)(t)],0..subs(pars,tf),color=blue
  );
  end do:
> b1:=odeplot(sol,[t,y1(t)],0..1,color=red):
  b2:=odeplot(sol,[t,z1(t)],0..1,color=blue):
```

```
> display(b1,b2,axes=boxed);
```




```
> display(seq(a||x,x=1..NODE+NAE),axes=boxed);
```




End Example 2

```
>
```

**Example 3**

```
> restart;
> with(plots):
> eq1:=diff(y1(t),t)^2+diff(y1(t),t)*(y1(t)+1)+y1(t)=cos(diff(y1(t),
  t));
```

$$eq1 := \left(\frac{d}{dt}y1(t)\right)^2 + \left(\frac{d}{dt}y1(t)\right)(y1(t)+1) + y1(t) = \cos\left(\frac{d}{dt}y1(t)\right)$$

```
> solx:=dsolve({eq1,y1(0)=0},numeric):
Error, (in dsolve/numeric/make_proc) Could not convert to an explicit first order
system due to 'RootOf'

> eqode:=[diff(y1(t),t)=z1(t)];
```

$$\text{eqode} := \left[\frac{d}{dt}\,y1(t) = z1(t)\right]$$

```
> eqae:=[subs(eqode,eq1)];
```

$$\text{eqae} := [z1(t)^2 + z1(t)\,(y1(t)+1) + y1(t) = \cos(z1(t))]$$

```
> icodes:=[y1(0)=0.0];
```

$$\text{icodes} := [y1(0) = 0.]$$

```
> icaes:=[z1(0)=0.0];
```

$$\text{icaes} := [z1(0) = 0.]$$

```
> pars:=[mu=0.1,q=1000,tint=1,tf=4];
```

$$\text{pars} := [\mu = 0.1,\ q = 1000,\ \text{tint} = 1,\ \text{tf} = 4]$$

```
> Xexplicit:=2;
```

$$\text{Xexplicit} := 2$$

```
> ff:=subs(pars,1/2+1/2*tanh(q*(t-tint)));
```

$$\text{ff} := \frac{1}{2} + \frac{1}{2}\tanh(1000\,t - 1000)$$

```
> NODE:=nops(eqode);NAE:=nops(eqae);
```

$$\text{NODE} := 1$$

$$\text{NAE} := 1$$

```
> for XX from 1 to NODE do
  EQODE||XX:=lhs(eqode[XX])=rhs(eqode[XX])*ff:
  end do;
```

$$\text{EQODE1} := \frac{d}{dt}\,y1(t) = z1(t)\left(\frac{1}{2} + \frac{1}{2}\tanh(1000\,t - 1000)\right)$$

```
> for XX from 1 to NAE do
  EQAE||XX:=subs(pars,-mu*(diff(rhs(eqae[XX])-lhs(eqae[XX]),t))=rhs(
  eqae[XX])-lhs(eqae[XX]));
  end do;
```

$$\text{EQAE1} := 0.1\sin(z1(t))\left(\frac{d}{dt}\,z1(t)\right) + 0.2\,z1(t)\left(\frac{d}{dt}\,z1(t)\right) + 0.1\left(\frac{d}{dt}\,z1(t)\right)(y1(t)+1)$$
$$+ 0.1\,z1(t)\left(\frac{d}{dt}\,y1(t)\right) + 0.1\left(\frac{d}{dt}\,y1(t)\right) = \cos(z1(t)) - z1(t)^2 - z1(t)\,(y1(t)+1) - y1(t)$$

```
>
> Dvars1:={seq(diff(z||x(t),t)=D||x,x=1..NAE)};
```

$$\text{Dvars1} := \left\{\frac{d}{dt}\,z1(t) = D1\right\}$$

```
> Dvars2:={seq(rhs(Dvars1[x])=lhs(Dvars1[x]),x=1..NAE)};
```

$$\text{Dvars2} := \left\{D1 = \frac{d}{dt}\,z1(t)\right\}$$

```
> icsn:=seq(subs(y||x(0)=y||x(t),icodes[x]),x=1..NODE),seq(subs(z||x
  (0)=z||x(t),icaes[x]),x=1..NAE);
```

$$icsn := y1(t) = 0., z1(t) = 0.$$

```
> for j from 1 to NAE do
> EQAEX||j:=subs(Dvars1,eqode,icsn,Dvars2,lhs(EQAE||j))=rhs(EQAE||j)
  ;
> end do;
```

EQAEX1 :=

$$0.1\sin(0.)\left(\frac{d}{dt}z1(t)\right)+0.1\left(\frac{d}{dt}z1(t)\right)=\cos(z1(t))-z1(t)^2-z1(t)\,(y1(t)+1)-y1(t)$$

```
> Sys:={seq(EQODE||x,x=1..NODE),seq(EQAEX||x,x=1..NAE),seq(icodes[x]
  ,x=1..NODE),seq(icaes[x],x=1..NAE)};
```

Sys := $\{0.1\left(\frac{d}{dt}z1(t)\right)=\cos(z1(t))-z1(t)^2-z1(t)\,(y1(t)+1)-y1(t),\ y1(0)=0.,\ z1(0)=0.,$

$$\frac{d}{dt}y1(t)=z1(t)\left(\frac{1}{2}+\frac{1}{2}\tanh(1000\,t-1000)\right)\}$$

```
> if Xexplicit=1 then
> sol:=dsolve(Sys,numeric):
> else
> sol:=dsolve(Sys,numeric,stiff=true,implicit=true):
> end if:
>
> for XX from 1 to NODE do
  a||XX:=odeplot(sol,[t,y||XX(t)],0..subs(pars,tf),color=red);
  end do:
> for XX from NODE+1 to NODE+NAE do
  a||XX:=odeplot(sol,[t,z||(XX-NODE)(t)],0..subs(pars,tf),color=blue
  );
  end do:
> display(seq(a||x,x=1..NODE+NAE),axes=boxed);
```




End Example 3

```
>
```

**Example 4**

```
> restart;
> with(plots):
> N:=11:h:=1/(N+1):
> for i from 2 to N+1 do
  eq1[i]:=diff(y||i(t),t)=(y||(i+1)(t)-2*y||i(t)+y||(i-1)(t))/h^2-y|
  |i(t)*(1+z||i(t));od:
> for i from 2 to N+1 do
  eq2[i]:=0=(z||(i+1)(t)-2*z||i(t)+z||(i-1)(t))/h^2-(1-y||i(t)^2)*(e
  xp(-z||i(t)));od:
> eq1[1]:=(3*y1(t)-4*y2(t)+y3(t))/(2*h)=0:
  eq1[N+2]:=y||(N+2)(t)-1=0:
> eq2[1]:=(3*z1(t)-4*z2(t)+1*z3(t))/(2*h)=0:
  eq2[N+2]:=z||(N+2)(t)=0:
> eq1[1]:=subs(y1(t)=z||(N+3)(t),eq1[1]):
> eq1[N+2]:=subs(y||(N+2)(t)=z||(N+4)(t),eq1[N+2]):
> eqode:=[seq(subs(y1(t)=z||(N+3)(t),y||(N+2)(t)=z||(N+4)(t),eq1[i])
  ,i=2..N+1)]:
> eqae:=[eq1[1],eq1[N+2],seq(eq2[i],i=1..N+2)]:
> icodes:=[seq(y||j(0)=1,j=2..N+1)]:
> icaes:=[seq(z||j(0)=0,j=1..N+2),z||(N+3)(0)=1,z||(N+4)(0)=1]:
> pars:=[mu=0.00001,q=1000,tint=1,tf=2]:
> Xexplicit:=2:
> ff:=subs(pars,1/2+1/2*tanh(q*(t-tint))):
> NODE:=nops(eqode):NAE:=nops(eqae):
> for XX from 1 to NODE do
> EQODE||XX:=lhs(eqode[XX])=rhs(eqode[XX])*ff: end do:
> for XX from 1 to NAE do
> EQAE||XX:=subs(pars,-mu*(diff(rhs(eqae[XX])-lhs(eqae[XX]),t))=rhs(
  eqae[XX])-lhs(eqae[XX])); end do:
> Dvars1:={seq(diff(z||x(t),t)=D||x,x=1..NAE)}:
> Dvars2:={seq(rhs(Dvars1[x])=lhs(Dvars1[x]),x=1..NAE)}:
> icsn:=seq(subs(y||x(0)=y||x(t),icodes[x]),x=1..NODE),seq(subs(z||x
  (0)=z||x(t),icaes[x]),x=1..NAE):
> for j from 1 to NAE do
> EQAEX||j:=subs(Dvars1,eqode,icsn,Dvars2,lhs(EQAE||j))=rhs(EQAE||j)
  :
> end do:
> Sys:={seq(EQODE||x,x=1..NODE),seq(EQAEX||x,x=1..NAE),seq(icodes[x]
  ,x=1..NODE),seq(icaes[x],x=1..NAE)}:
> if Xexplicit=1 then
> sol:=dsolve(Sys,numeric,maxfun=0):
```

```
> else
> sol:=dsolve(Sys,numeric,stiff=true,implicit=true,maxfun=0):
> end if:
>
> for XX from 1 to NODE do
> a||XX:=odeplot(sol,[t,y||(XX+1)(t)],1..subs(pars,tf),color=red):
  end do:
> for XX from NODE+1 to NODE+NAE do
> a||XX:=odeplot(sol,[t,z||(XX-NODE)(t)],1..subs(pars,tf),color=blue
  ): end do:
> display(seq(a||x,x=1..NODE),a||(NODE+NAE-1),a||(NODE+NAE),axes=box
  ed);
```




End of Example 4

>

Sometimes the parameters of the switch function and perturbation need to be tuned to obtain propoer convergence. Below is Example 1 shown for several cases using the 'parameters' option in Maple's dsolve to compare how tuning parameters affects the solution

```
> restart:
> with(plots):
> eqode:=[diff(y1(t),t)=-y1(t)^2+z1(t)]:
  eqae:=[cos(y1(t))-z1(t)^0.5=0]:
> icodes:=[y1(0)=0.25]: icaes:=[z1(0)=0.8]:
> pars:=[tf=5]:
> Xexplicit:=2;
```

Xexplicit := 2

```
> ff:=subs(pars,1/2+1/2*tanh(q*(t-tint))):
> NODE:=nops(eqode):NAE:=nops(eqae):
> for XX from 1 to NODE do
  EQODE||XX:=lhs(eqode[XX])=rhs(eqode[XX])*ff:
```

```
  end do:
> for XX from 1 to NAE do
  EQAE||XX:=subs(pars,-mu*(diff(rhs(eqae[XX])-lhs(eqae[XX]),t))=rhs(
  eqae[XX])-lhs(eqae[XX]));
  end do:
>
> Dvars1:={seq(diff(z||x(t),t)=D||x,x=1..NAE)}:
> Dvars2:={seq(rhs(Dvars1[x])=lhs(Dvars1[x]),x=1..NAE)}:
> icsn:=seq(subs(y||x(0)=y||x(t),icodes[x]),x=1..NODE),seq(subs(z||x
  (0)=z||x(t),icaes[x]),x=1..NAE):
> for j from 1 to NAE do
> EQAEX||j:=subs(Dvars1,eqode,icsn,Dvars2,lhs(EQAE||j))=rhs(EQAE||j)
  :
> end do:
> Sys:={seq(EQODE||x,x=1..NODE),seq(EQAEX||x,x=1..NAE),seq(icodes[x]
  ,x=1..NODE),seq(icaes[x],x=1..NAE)}:
> if Xexplicit=1 then
> sol:=dsolve(Sys,numeric,'parameters'=[mu,q,tint],maxfun=0):
> else
> sol:=dsolve(Sys,numeric,'parameters'=[mu,q,tint],stiff=true,implic
  it=true):
> end if:
>
> sol('parameters'=[0.1,1000,1]):
> plot1:=odeplot(sol,[t-1,y1(t)],0..4,color=red):
  plot2:=odeplot(sol,[t-1,z1(t)],0..4,color=blue):
> sol('parameters'=[0.001,10,1]):
> plot3:=odeplot(sol,[t-1,y1(t)],0..4,color=red,linestyle=dot):
  plot4:=odeplot(sol,[t-1,z1(t)],0..4,color=blue,linestyle=dot):
> display(plot1,plot2,plot3,plot4,axes=boxed);
```




In general, one has to decrease mu, and increase q and tint until convergence (example at t=3)

```
> sol('parameters'=[0.001,10,1]):sol(3+1);
```

$$[t = 4., y1(t) = 0.738587929442734, z1(t) = 0.546472878850096]$$

```
> sol('parameters'=[0.0001,100,10]):sol(3+10);
```

$$[t = 13., y1(t) = 0.738684397167344, z1(t) = 0.546618936273638]$$

```
> sol('parameters'=[0.00001,1000,20]):sol(3+20);
```

$$[t = 23., y1(t) = 0.738694113087217, z1(t) = 0.546633473784526]$$

```
>
```

The results have converged to 4 digits after the decimal. Of course, absolute and relative tolerances of the solvers can be modified if needed